

STAT 5200 Handout #12: Factorial Design Without Replication (Ch. 8)

Example: An experiment is carried out to evaluate the effects of three factors on the electrical conductivity of soil. The factors in the experiment are:

- **Soil Type.** This factor has three levels: Clay, Loam, and Sand.
- **Salinity.** This factor has three levels: 2, 8, and 16.
- **Water:** the amount of water added to the soil samples, in ml. This factor also has three levels: 0, 5, and 15.

In a completely random order, one sample for each combination of factor levels is constructed and a single measurement of the electrical conductivity (**EC**) of the sample is made. This experiment is a $3 \times 3 \times 3$ factorial design with 1 replicate per combination of factor levels.

```
/* STAT 5200
   factorial design (unreplicated)
   electrical conductivity data
   -- a 3 x 3 x 3 factorial design with 1 replicate
       per factor level combination
*/

/* Read in data */
data electric;
input Soil $ Salinity Water EC @@;
cards;
Sand 2 0 0.60 Sand 2 5 1.69 Sand 2 15 3.47
Sand 8 0 0.05 Sand 8 5 0.11 Sand 8 15 0.06
Sand 16 0 0.07 Sand 16 5 0.08 Sand 16 15 0.22
Loam 2 0 0.98 Loam 2 5 2.21 Loam 2 15 5.68
Loam 8 0 0.15 Loam 8 5 0.23 Loam 8 15 0.40
Loam 16 0 0.07 Loam 16 5 0.23 Loam 16 15 0.43
Clay 2 0 1.37 Clay 2 5 3.31 Clay 2 15 5.74
Clay 8 0 0.72 Clay 8 5 0.78 Clay 8 15 2.10
Clay 16 0 0.40 Clay 16 5 0.72 Clay 16 15 1.95
;
run;
```

```

/* Fit model */
proc glm data=electric;
  class Soil Salinity Water;
  model EC = Soil | Salinity | Water;
  title1 'Full model with 3-way interaction';
run;

```

Full model with 3-way interaction					
The GLM Procedure					
Dependent Variable: EC					
Source	DF	Sum of Squares	Mean Square	F Value	Pr > F
Model	26	67.35551852	2.59059687	.	.
Error	0	0.00000000	.		
Corrected Total	26	67.35551852			

R-Square	Coeff Var	Root MSE	EC Mean
1.000000	.	.	1.252593

Source	DF	Type III SS	Mean Square	F Value	Pr > F
Soil	2	6.54120741	3.27060370	.	.
Salinity	2	31.64302963	15.82151481	.	.
Soil*Salinity	4	0.83388148	0.20847037	.	.
Water	2	14.19956296	7.09978148	.	.
Soil*Water	4	1.64474815	0.41118704	.	.
Salinity*Water	4	11.51659259	2.87914815	.	.
Soil*Salinity*Water	8	0.97649630	0.12206204	.	.

```

/* Account for no replicates */
proc glm data=electric plots=diagnostic;
  class Soil Salinity Water;
  model EC = Soil | Salinity | Water @ 2;
  title 'Restrict to 2-way interactions';
run;

```

up to order 2

Restrict to 2-way interactions


The GLM Procedure

Dependent Variable: EC

Source	DF	Sum of Squares	Mean Square	F Value	Pr > F
Model	18	66.37902222	3.68772346	30.21	<.0001
Error	8	0.97649630	0.12206204		
Corrected Total	26	67.35551852			

R-Square	Coeff Var	Root MSE	EC Mean
0.985502	27.89205	0.349374	1.252593


Source	DF	Type III SS	Mean Square	F Value	Pr > F
Soil	2	6.54120741	3.27060370	26.79	0.0003
Salinity	2	31.64302963	15.82151481	129.62	<.0001
Soil*Salinity	4	0.83388148	0.20847037	1.71	0.2406
Water	2	14.19956296	7.09978148	58.17	<.0001
Soil*Water	4	1.64474815	0.41118704	3.37	0.0676
Salinity*Water	4	11.51659259	2.87914815	23.59	0.0002



```

/* Consider transformation */
proc transreg data=electric;
  model boxcox(EC / lambda=0 to 1 by 0.05)
    = class(Soil Salinity Water
            Soil*Salinity Soil*Water Salinity*Water);
  title 'Box-Cox on response';
run;

```



```

/* Justify log (lambda=0) even though outside
   Box-Cox interval -- log-likelihood fairly flat,
   and residuals look great (next page) */

```

```




/* Use transformed EC; account for no replicates */
data electric; set electric;
  new_EC = log(EC);
proc glimmix data=electric plots=residualpanel;
  class Soil Salinity Water;
  model new_EC = Soil | Salinity | Water @ 2;
  lsmeans Soil*Salinity / pdiff adjust=tukey lines
    plot=mean(sliceby=Soil join);
  lsmeans Salinity*Soil / plot=mean(sliceby=Salinity join);
  title '(Transformed) Restrict to 2-way interactions';
run;
/* Recall: SLICEBY tells which variable to use
   as plotting characters */

```

(Transformed) Restrict to 2-way interactions

The GLIMMIX Procedure

Type III Tests of Fixed Effects				
Effect	Num DF	Den DF	F Value	Pr > F
Soil	2	8	78.02	<.0001
Salinity	2	8	143.05	<.0001
Soil*Salinity	4	8	9.37	0.0041
Water	2	8	39.92	<.0001
Soil*Water	4	8	0.75	0.5865
Salinity*Water	4	8	2.00	0.1877



**Tukey Grouping for Soil*Salinity
Least Squares Means (Alpha=0.05)**

**LS-means with the same
letter are not significantly
different.**

Soil	Salinity	Estimate		
Clay	2	1.0864		A
				A
Loam	2	0.8366	B	A
			B	A
Sand	2	0.4194	B	A
			B	
Clay	8	0.05499	B	
			B	
Clay	16	-0.1923	B	
Loam	8	-1.4277		C
				C
Loam	16	-1.6576	D	C
			D	C
Sand	16	-2.2330	D	C
			D	
Sand	8	-2.6721	D	